Imperial College London

Coursework 1

Imperial College London

Department of Mathematics

Formalising mathematics

[redacted]

Date: February 4, 2022

1 Kernel of a group homomorphism is a normal subgroup of the domain

In this first part we will prove that the kernel of a group homomorphism is a normal subgroup of the domain of the homomorphism.

The mathematical proof for it is rather short.

Theorem 1
Let ϕ be a group homomorphism, then the kernel of ϕ is a normal subgroup of the domain of ϕ.

$$\ker(\phi) \triangleleft Dom(\phi) \tag{1}$$

Proof 1
Let $\phi : G \rightarrow H$, where the identities of G and H are e_G and e_H respectively.

The kernel $\ker(\phi)$ is a subgroup of G.

Let $k \in \ker(\phi), x \in G$. Then:

$$\begin{aligned}\phi(xkx^{-1}) &= \phi(x)\phi(k)\phi(x^{-1}) \\ &= \phi(x)e_k\phi(x^{-1}) \\ &= \phi(x)\phi(x^{-1}) \\ &= \phi(xx^{-1}) \\ &= \phi(e_G) \\ &= e_H\end{aligned} \tag{2}$$

Therefore $xkx^{-1} \in \ker(\phi)$, and this is true for all $k \in \ker(\phi), x \in G$.

As the definition of a normal subgroup is that it contains all conjugate elements, it follows that the kernel is a normal subgroup of its domain.

This statement is already proven in Lean's mathlib by using `monoid\_hom.normal\_ker`, however we will prove this ourselves using mathlib's groups and subgroups.

To help us we introduce a couple of helper lemmas:

```
variables {G H : Type} [group G] [group H]

lemma map_mul_2 {x y : G}{φ : G →* H} :
  φ(x * y * x⁻¹) = φ(x) * φ(y) * φ(x)⁻¹ :=
begin
    -- rewrite twice to get an obvious identity
  rw map_mul,
  rw map_mul,
end
```

```
lemma x_in_kernel_is_identity {k : G}{φ : G →* H}(hk : k ∈ φ.ker) :
  φ(k) = 1 :=
begin
  exact hk,
end
```

The proofs for these lemmas are just simple rewriting to get a trivial equality and a definition equivalence. We introduce them so that we can rewrite these statements in the lemma which does most of the heavy lifting for the final proof.

The main step of the proof is to show that $\phi(xkx^{-1}) = e_H$, in Lean we create a new lemma which does just that:

```
lemma conjugating_k_in_kernel_by_x_is_identity
  {k : G}{φ : G →* H}(hk : k ∈ φ.ker){x : G} : φ(x * k * x⁻¹) = 1 :=
begin
  -- rewrite our hypothesis to expand to multiplication by function
  rw map_mul_2,
  -- kernel element goes to the identity
  rw x_in_kernel_is_identity hk,
  -- Remove the multiplication by 1
  rw mul_one,
  -- bring back the multiplication by maps to multiplication within a map
  rw ← map_mul,
  -- cancel x * x⁻¹
  rw mul_right_inv,
  -- identity maps to identity
  rw map_one,
end
```

Once more the proof is rewriting the statements, this time to reduce it to the identity map.

We write one final lemma, where we need to introduce the ϕ homomorphism so that Lean can apply this lemma in the final proof, where the proof is just applying the previous lemma and using the definition of what it means to be in the kernel:

```
variable {φ : G →* H}
lemma conjugating_kernel_by_x_is_in_kernel
  {k : G}(hk : k ∈ φ.ker){x : G} : x * k * x⁻¹ ∈ φ.ker :=
begin
    apply conjugating_k_in_kernel_by_x_is_identity,
    exact hk,
end
```

The theorem is finally stated and proved by changing the goal from being a normal subgroup, to what makes a group normal. And then applying the previous lemma we wrote.

```
theorem kernel_is_normal_subgroup_of_domain {φ : G →* H} :
  subgroup.normal (φ.ker) :=
begin
  -- Change the goal to be the hypothesis of what a normal group is
  apply subgroup.normal.mk,
  apply conjugating_kernel_by_x_is_in_kernel,
end
```

2 Preimage of a normal subgroup is normal

In this second part we prove that the preimage of a normal subgroup under a group homomorphism is normal.

Theorem 2
Let $\phi : G \to H$, if I is a normal subgroup of H then $P = \phi^{-1}(H)$ is a normal subgroup of G.

Proof 2
$\forall x \in G, \forall y \in P$ we have:

$$\phi(xyx^{-1}) = \phi(x)\phi(y)\phi(x)^{-1} \tag{3}$$

as ϕ is a group homomorphism. As $\phi(x) \in H, \phi(y) \in I$ and I is a normal subgroup, this means $\phi(x)\phi(y)\phi(x)^{-1} \in I$.

By the definition of P, we have that $xyx^{-1} \in P$, proving P is a normal subgroup.

The Lean code to prove the theorem is as follows:

```
theorem preimage_of_normal_subgroup_is_normal
  {φ : G →* H} {I : subgroup H}(hn : subgroup.normal(I)) :
    subgroup.normal (I.comap φ) :=
begin
  -- Change the goal to be the hypothesis of what a normal group is
  apply subgroup.normal.mk,
  -- y an element of G
  intro y,
  -- hypothesis that y is in the preimage of φ
  intro hyInPreim,
  -- x any element of G
  intro x,
  --Simplifies the goal from x * y * x⁻¹ ∈ subgroup.comap φ I
  -- to φ(x) * φ(y) * φ(x)⁻¹ ∈ I,
  simp,
  -- Apply the conjecture that the image I is a normal subgroup
  apply hn.conj_mem,
  -- φ(y) is in the image, thus proving the claim
  exact hyInPreim,
end
```